unittest
{
	assert( [ "1 1 2 2" ].parse.expand.solve == 14 );
	assert( [ "314 159 2653 589" ].parse.expand.solve == 602215194 );
}



import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto rcrc = input.front.split.to!( long[] );
	return tuple( rcrc[ 0 ], rcrc[ 1 ], rcrc[ 2 ], rcrc[ 3 ] );
}

auto solve( long r1, long c1, long r2, long c2 )
{
	// 二項係数の二次元累積和表は、2つの一次元累積和表で代替できる
	
	// 範囲の一つ外が欲しいので減らす
	r1--;
	c1--;
	
	// 累積和表を作成
	auto r1s = solve2( r1, c2 );
	auto r2s = solve2( r2, c2 );
	
	// 範囲内の総和を計算
	return r2s[ c2 ] - r2s[ c1 ] - r1s[ c2 ] + r1s[ c1 ];
}

auto solve2( long r, long c )
{
	auto ts = new ModWorld!()[ 2 + c ];
	foreach( i; 1L .. ts.length.signed )
	{
		ts[ i ] = ts[ i - 1 ] + ( r + i ).combination( i );
	}
	ts.popFront;
	return ts;
}

// モジュラ整数型（n mod M）
struct ModWorld( long M = 10L ^^ 9 + 7 )
{
	long n;
	
	// 演算子オーバーロード
	void opUnary( string OP )() if( OP == "++" ){ this.n = ( this.n + 1 ) % M; }
	void opUnary( string OP )() if( OP == "--" ){ this.n = ( this.n + M - 1 ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "+"  ){ this.n = ( this.n + r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "-"  ){ this.n = ( this.n + M - r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "*"  ){ this.n = ( this.n * r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "/"  ){ this *= r ^^ ModWorld( M - 2 ); }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "^^" )
	{
		auto a = this.n;
		this.n = 1L;
		for( auto i = r.n; 0 < i; i >>= 1 )
		{
			if( i & 1 ) this.n = this.n * a % M;
			a = a ^^ 2 % M;
		}
	}
	auto opBinary( string OP )( ModWorld r )
	{
		auto t = this;
		t.opOpAssign!OP( r );
		return t;
	}
	
	// long型対応
	void opAssign( long r ){ this.n = r; }
	void opOpAssign( string OP )( long r ){ this.opOpAssign!OP( ModWorld( r ) ); }
	auto opBinary( string OP )( long r ){ return this.opBinary!OP( ModWorld( r ) ); }
	auto opBinaryRight( string OP )( long l ){ return ModWorld( l ).opBinary!OP( this ); }
	auto opEquals( long r ){ return this.n == r; }
	
	// format用出力
	string toString()
	{
		return this.n.to!string;
	}
}

// 階乗
auto factorial( long M = 10L ^^ 9 + 7 )( long x )
{
	static memo = [ ModWorld!M( 1L ) ];
	if( memo.length <= x )
	{
		auto i = memo.length.signed;
		memo.length = x + 1;
		for( ; i <= x; i++ ) memo[ i ] = memo[ i - 1 ] * i;
	}
	return memo[ x ];
}

// 組み合わせ数
auto combination( long M = 10L ^^ 9 + 7 )( long n, long k )
{
	if( n < k ) return ModWorld!M( 0L );
	else        return n.factorial!M / ( k.factorial!M * ( n - k ).factorial!M );
}
